Appendix C: Software demonstration for Killing tensors of rank 2.

In this demonstration, we will illustrate the utility of the programs with the following names: MaxKT, BundleLift, HauserTractorConnection, getHauserKT2, and HauserTractorLift2.

First, we will read in the file which contains the programs we have written and also load in the required packages.

```
> read "TractorPrograms.txt";
  with(DifferentialGeometry):
  with(Tensor):
```

Now, we will initialize a 2-dimensional manifold.

```
> DGEnvironment[Coordinate]([u,v], M);
```

$$\text{Manifold: } M \tag{1}$$

Next, we will define a simple metric.

```
M > g := evalDG(u^p * (du &t du + dv &t dv) );
```

$$g := u^p\, du \otimes du + u^p\, dv \otimes dv \tag{2}$$

We need to know the required size of the fibers of the tractor bundle. We find this to be 6:

```
M > MaxKT(2, 2);
```

$$6 \tag{3}$$

Now we initialize the required environments.

```
M > DGEnvironment[VectorSpace](6, V);
```

$$\text{Vector Space: } V \tag{4}$$

```
V > DGEnvironment[VectorBundle](M, V, N);
```

$$\text{Vector Bundle: } N \tag{5}$$

Now that the vector bundle has been initialized, it is convenient to redefine the metric on this bundle:

```
N > G := BundleLift(g, N);
```

$$G := u^p\, du \otimes du + u^p\, dv \otimes dv \tag{6}$$

We will also need the Christoffel symbols:

```
N > Gamma := Christoffel(G);
```

$$\Gamma := \nabla_{\partial_u}\partial_u = \frac{p}{2\,u}\,\partial_u, \nabla_{\partial_u}\partial_v = \frac{p}{2\,u}\,\partial_v, \nabla_{\partial_v}\partial_u = \frac{p}{2\,u}\,\partial_v, \nabla_{\partial_v}\partial_v = -\frac{p}{2\,u}\,\partial_u \tag{7}$$

Now we can compute the tractor connection:

```
N > C := HauserTractorConnection(Gamma, N);
```

$$\begin{aligned} C := {}& \nabla_{\partial_u} E1 = -\frac{p}{u}\,E1 + \frac{3\,p}{4\,u^2}\,E5 - \frac{3\,p\,(2+p)}{4\,u^3}\,E6, \nabla_{\partial_u} E2 = -\frac{p}{u}\,E2 - \frac{3\,p}{2\,u^2}\,E4 \\ &, \nabla_{\partial_u} E3 = -\frac{p}{u}\,E3 - \frac{3\,p}{4\,u^2}\,E5 + \frac{3\,p\,(2+p)}{4\,u^3}\,E6, \nabla_{\partial_u} E4 = \left(\frac{1}{3}\right) E2 - \frac{3\,p}{2\,u}\,E4 \\ &, \nabla_{\partial_u} E5 = \left(\frac{2}{3}\right) E3 - \frac{3\,p}{2\,u}\,E5 + \frac{3\,p}{2\,u^2}\,E6, \nabla_{\partial_u} E6 = -E5 - \frac{2\,p}{u}\,E6, \nabla_{\partial_v} E1 = \frac{p}{2\,u}\,E2 \\ &+ \frac{3\,p}{4\,u^2}\,E4, \nabla_{\partial_v} E2 = -\frac{p}{u}\,E1 + \frac{p}{u}\,E3 + \frac{3\,p}{2\,u^2}\,E5 - \frac{3\,p\,(2+p)}{2\,u^3}\,E6, \nabla_{\partial_v} E3 = \\ &-\frac{p}{2\,u}\,E2 - \frac{3\,p}{4\,u^2}\,E4, \nabla_{\partial_v} E4 = -\left(\frac{2}{3}\right) E1 + \frac{p}{2\,u}\,E5 - \frac{3\,p}{2\,u^2}\,E6, \nabla_{\partial_v} E5 = -\left(\frac{1}{3}\right) E2 \\ &- \frac{p}{2\,u}\,E4, \nabla_{\partial_v} E6 = E4 \end{aligned} \tag{8}$$

We can also represent this connection as a matrix of 1-forms:

```
N > convert(C, DGMatrix);
```

$$\begin{aligned} \Bigg[& \left[-\frac{p}{u}\,du, -\frac{p}{u}\,dv, 0\,du, -\left(\frac{2}{3}\right) dv, 0\,du, 0\,du \right], \\ & \left[\frac{p}{2\,u}\,dv, -\frac{p}{u}\,du, -\frac{p}{2\,u}\,dv, \left(\frac{1}{3}\right) du, -\left(\frac{1}{3}\right) dv, 0\,du \right], \\ & \left[0\,du, \frac{p}{u}\,dv, -\frac{p}{u}\,du, 0\,du, \left(\frac{2}{3}\right) du, 0\,du \right], \\ & \left[\frac{3\,p}{4\,u^2}\,dv, -\frac{3\,p}{2\,u^2}\,du, -\frac{3\,p}{4\,u^2}\,dv, -\frac{3\,p}{2\,u}\,du, -\frac{p}{2\,u}\,dv, dv \right], \\ & \left[\frac{3\,p}{4\,u^2}\,du, \frac{3\,p}{2\,u^2}\,dv, -\frac{3\,p}{4\,u^2}\,du, \frac{p}{2\,u}\,dv, -\frac{3\,p}{2\,u}\,du, -du \right], \\ & \left[-\frac{3\,p\,(2+p)}{4\,u^3}\,du, -\frac{3\,p\,(2+p)}{2\,u^3}\,dv, \frac{3\,p\,(2+p)}{4\,u^3}\,du, -\frac{3\,p}{2\,u^2}\,dv, \frac{3\,p}{2\,u^2}\,du, \right. \\ & \left. -\frac{2\,p}{u}\,du \right] \Bigg] \end{aligned} \tag{9}$$

We now compute the Curvature tensor for the tractor connection.

```
N > K := CurvatureTensor(C);
```

$$K := \frac{3\,p\,(2+p)\,(3+2\,p)}{2\,u^4}\,E6 \otimes \Theta 2 \otimes du \otimes dv - \frac{3\,p\,(2+p)\,(3+2\,p)}{2\,u^4}\,E6 \otimes \Theta 2 \otimes dv \otimes du + \frac{5\,p\,(2+p)}{2\,u^3}\,E6 \otimes \Theta 4 \otimes du \otimes dv - \frac{5\,p\,(2+p)}{2\,u^3}\,E6 \otimes \Theta 4 \otimes dv \otimes du \tag{10}$$

It is interesting to note that p=0 and p=-2 are the only values of p for which the curvature tensor vanishes identically.

We can represent this tensor as a collection of (1,1) tensors by contracting the curvature tensor with the coordinate vectors of the base manifold. In our 2-dimensional case, there is only one such (1,1) tensor, but there may be others in higher dimensions.

```
N > k1 := ContractIndices(K, evalDG(D_u &t D_v), [[3,1],[4,2]] );
```

$$k1 := \frac{3\,p\,(2+p)\,(3+2\,p)}{2\,u^4}\,E6 \otimes \Theta 2 + \frac{5\,p\,(2+p)}{2\,u^3}\,E6 \otimes \Theta 4 \tag{11}$$

We can think of this as a matrix

```
N > K1 := convert(k1, DGMatrix);
```

$$K1 := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{3}{2}\,\frac{p\,(2+p)\,(3+2\,p)}{u^4} & 0 & \frac{5}{2}\,\frac{p\,(2+p)}{u^3} & 0 & 0 \end{bmatrix} \tag{12}$$

and subsequently find the basis for the 0th order reduced tractor bundle:

```
N > IT0 := LieAlgebras:-InvariantTensors([K1], [seq(E||i,i=1..6)] ) ;
```

$$IT0 := \left[E1,\ E2 - \frac{3\,(3+2\,p)}{5\,u}\,E4,\ E3,\ E5,\ E6\right] \tag{13}$$

How many are there?

```
N > nops(IT0);
```

$$5 \tag{14}$$

This calculation is assuming that p is not 0 or 2: caution is advised when examining metrics with unknown constants as exponents. Assuming that p is not 0 or 2, there are a maximum of 5 Killing tensors of the metric g..

We will now try to get a tighter upper bound. We begin by differentiating the curvature tensor.

```
N > dK1 := CovariantDerivative(k1, C);
```

$$\begin{aligned} dK1 := {} & \frac{3\,p\,(2+p)\,(3+2\,p)}{2\,u^4}\, E4 \otimes \Theta 2 \otimes dv + \frac{5\,p\,(2+p)}{2\,u^3}\, E4 \otimes \Theta 4 \otimes dv \\ & - \frac{3\,p\,(2+p)\,(3+2\,p)}{2\,u^4}\, E5 \otimes \Theta 2 \otimes du - \frac{5\,p\,(2+p)}{2\,u^3}\, E5 \otimes \Theta 4 \otimes du \\ & - \frac{3\,p^2\,(2+p)\,(11+4\,p)}{8\,u^5}\, E6 \otimes \Theta 1 \otimes dv - \frac{3\,p\,(2+p)\,\left(4\,p^2+17\,p+24\right)}{4\,u^5}\, E6 \\ & \otimes \Theta 2 \otimes du + \frac{3\,p^2\,(2+p)\,(11+4\,p)}{8\,u^5}\, E6 \otimes \Theta 3 \otimes dv - \frac{9\,p\,(2+p)\,(4+p)}{4\,u^4}\, E6 \\ & \otimes \Theta 4 \otimes du + \frac{3\,p\,(2+p)\,(2+3\,p)}{4\,u^4}\, E6 \otimes \Theta 5 \otimes dv - \frac{5\,p\,(2+p)}{2\,u^3}\, E6 \otimes \Theta 6 \otimes dv \end{aligned} \tag{15}$$

We now generate a set of (1,1) tensors by contraction:

```
N > D1K1 := ContractIndices(dK1, D_u, [[3,1]]);
```

$$\begin{aligned} D1K1 := {} & -\frac{3\,p\,(2+p)\,(3+2\,p)}{2\,u^4}\, E5 \otimes \Theta 2 - \frac{5\,p\,(2+p)}{2\,u^3}\, E5 \otimes \Theta 4 \\ & - \frac{3\,p\,(2+p)\,\left(4\,p^2+17\,p+24\right)}{4\,u^5}\, E6 \otimes \Theta 2 - \frac{9\,p\,(2+p)\,(4+p)}{4\,u^4}\, E6 \otimes \Theta 4 \end{aligned} \tag{16}$$

```
N > D2K1 := ContractIndices(dK1, D_v, [[3,1]]);
```

$$\begin{aligned} D2K1 := {} & \frac{3\,p\,(2+p)\,(3+2\,p)}{2\,u^4}\, E4 \otimes \Theta 2 + \frac{5\,p\,(2+p)}{2\,u^3}\, E4 \otimes \Theta 4 \\ & - \frac{3\,p^2\,(2+p)\,(11+4\,p)}{8\,u^5}\, E6 \otimes \Theta 1 + \frac{3\,p^2\,(2+p)\,(11+4\,p)}{8\,u^5}\, E6 \otimes \Theta 3 \\ & + \frac{3\,p\,(2+p)\,(2+3\,p)}{4\,u^4}\, E6 \otimes \Theta 5 - \frac{5\,p\,(2+p)}{2\,u^3}\, E6 \otimes \Theta 6 \end{aligned} \tag{17}$$

Now we will convert them to matrices.

```
N > Md1k1 := convert(D1K1, DGMatrix);
```

$$Md1k1 := \left[\left[0, 0, 0, 0, 0, 0\right],\right. \tag{18}$$

$$\left[0, 0, 0, 0, 0, 0\right],$$

$$\left[0, 0, 0, 0, 0, 0\right],$$

$$\left[0, 0, 0, 0, 0, 0\right],$$

$$\left[0, -\frac{3}{2}\ \frac{p\,(2+p)\,(3+2\,p)}{u^4}, 0, -\frac{5}{2}\ \frac{p\,(2+p)}{u^3}, 0, 0\right],$$

$$\left.\left[0, -\frac{3}{4}\ \frac{p\,(2+p)\,\left(4\,p^2+17\,p+24\right)}{u^5}, 0, -\frac{9}{4}\ \frac{p\,(2+p)\,(4+p)}{u^4}, 0, 0\right]\right]$$

```
N > Md2k1 := convert(D2K1, DGMatrix);
```

$$Md2k1 := \left[\left[0, 0, 0, 0, 0, 0\right],\right. \tag{19}$$

$$\left[0, 0, 0, 0, 0, 0\right],$$

$$\left[0, 0, 0, 0, 0, 0\right],$$

$$\left[0, \frac{3}{2}\ \frac{p\,(2+p)\,(3+2\,p)}{u^4}, 0, \frac{5}{2}\ \frac{p\,(2+p)}{u^3}, 0, 0\right],$$

$$\left[0, 0, 0, 0, 0, 0\right],$$

$$\left[-\frac{3}{8}\ \frac{p^2\,(2+p)\,(11+4\,p)}{u^5}, 0, \frac{3}{8}\ \frac{p^2\,(2+p)\,(11+4\,p)}{u^5}, 0,\right.$$

$$\left.\left.\frac{3}{4}\ \frac{p\,(2+p)\,(2+3\,p)}{u^4}, -\frac{5}{2}\ \frac{p\,(2+p)}{u^3}\right]\right]$$

And now we can find a basis for the 1st order reduced tractor bundle:

```
N > IT1 := LieAlgebras:-InvariantTensors([K1,Md1k1,Md2k1], [seq
    (E||i,i=1..6)]);
```

$$IT1 := \left[E1 - \frac{3\,p\,(11+4\,p)}{20\,u^2}\ E6, E3 + \frac{3\,p\,(11+4\,p)}{20\,u^2}\ E6, E5 + \frac{3\,(2+3\,p)}{10\,u}\ E6\right] \tag{20}$$

```
N > nops(IT1);
```

$$3 \tag{21}$$

We are led to believe that there is a maximum of 3 Killing tensors. However, if p=1, we get 4 basis elements:

```
N > rIT1 := LieAlgebras:-InvariantTensors(eval([K1,Md1k1,Md2k1], p=1), [seq(E||i,i=1..6)]);
```

$$rIT1 := \left[E1 - \frac{9}{4\,u^2}\,E6,\ E2 - \frac{3}{u}\,E4,\ E3 + \frac{9}{4\,u^2}\,E6,\ E5 + \frac{3}{2\,u}\,E6\right] \quad \textbf{(22)}$$

```
N > nops(rIT1);
```

$$4 \quad \textbf{(23)}$$

Thus, it is not advisable to work with metrics which have unknown exponents, unless the exceptional values for those exponents are being sought.

Now let's see if we can get the Killing tensors explicitly for p=1. We begin by forming a function-coefficient linear combination of the basis elements of the 1st order reduced tractor bundle.

```
N > s1 := DGzip([seq(q||i(u,v), i=1..nops(rIT1))], rIT1, "plus");
```

$$s1 := q1(u,v)\,E1 + q2(u,v)\,E2 + q3(u,v)\,E3 - \frac{3\,q2(u,v)}{u}\,E4 + q4(u,v)\,E5 + \left(-\frac{9\,q1(u,v)}{4\,u^2} + \frac{9\,q3(u,v)}{4\,u^2} + \frac{3\,q4(u,v)}{2\,u}\right)E6 \quad \textbf{(24)}$$

Next, we take the covariant derivative.

```
N > s2 := CovariantDerivative(s1, eval(C, p=1) );
```

$$\begin{aligned} s2 := {} & \left(\frac{\partial}{\partial u}\,q1(u,v) - \frac{q1(u,v)}{u}\right) E1 \otimes du + \left(\frac{\partial}{\partial v}\,q1(u,v) + \frac{q2(u,v)}{u}\right) E1 \otimes dv + \\ & \left(\frac{\partial}{\partial u}\,q2(u,v) - \frac{2\,q2(u,v)}{u}\right) E2 \otimes du \\ & + \frac{-2\,q4(u,v)\,u + 6\left(\frac{\partial}{\partial v}\,q2(u,v)\right)u + 3\,q1(u,v) - 3\,q3(u,v)}{6\,u}\,E2 \otimes dv + \left(\frac{\partial}{\partial u}\right. \\ & \left. q3(u,v) - \frac{q3(u,v)}{u} + \frac{2\,q4(u,v)}{3}\right) E3 \otimes du + \left(\frac{\partial}{\partial v}\,q3(u,v) + \frac{q2(u,v)}{u}\right) E3 \otimes d \\ & v + \frac{3\left(-\left(\frac{\partial}{\partial u}\,q2(u,v)\right)u + 2\,q2(u,v)\right)}{u^2}\,E4 \otimes du \end{aligned} \quad \textbf{(25)}$$

$$- \frac{-2\, q4(u, v)\, u + 6 \left(\frac{\partial}{\partial v}\, q2(u, v)\right) u + 3\, q1(u, v) - 3\, q3(u, v)}{2\, u^2}\, E4 \otimes dv$$
$$+ \frac{\left(\frac{\partial}{\partial u}\, q4(u, v)\right) u^2 - 3\, q4(u, v)\, u + 3\, q1(u, v) - 3\, q3(u, v)}{u^2}\, E5 \otimes du + \frac{\partial}{\partial v}$$
$$q4(u, v)\, E5 \otimes dv + \frac{1}{4\, u^3} \left(3 \left(2 \left(\frac{\partial}{\partial u}\, q4(u, v)\right) u^2 - 4\, q4(u, v)\, u - 3 \left(\frac{\partial}{\partial u}\right.\right.\right.$$
$$\left.\left.\left. q1(u, v)\right) u + 3 \left(\frac{\partial}{\partial u}\, q3(u, v)\right) u + 9\, q1(u, v) - 9\, q3(u, v)\right)\right) E6 \otimes du$$
$$- \frac{3 \left(-2 \left(\frac{\partial}{\partial v}\, q4(u, v)\right) u + 3 \left(\frac{\partial}{\partial v}\, q1(u, v)\right) - 3 \left(\frac{\partial}{\partial v}\, q3(u, v)\right)\right)}{4\, u^2}\, E6 \otimes dv$$

Let's look at the equations we need to solve.

```
N > s3 := DGinformation(s2, "CoefficientSet");
```

$$s3 := \left\{ \frac{\left(\frac{\partial}{\partial u}\, q4(u, v)\right) u^2 - 3\, q4(u, v)\, u + 3\, q1(u, v) - 3\, q3(u, v)}{u^2}, \right. \qquad \textbf{(26)}$$
$$\frac{3 \left(-\left(\frac{\partial}{\partial u}\, q2(u, v)\right) u + 2\, q2(u, v)\right)}{u^2},$$
$$-\frac{3}{4}\, \frac{-2 \left(\frac{\partial}{\partial v}\, q4(u, v)\right) u + 3 \left(\frac{\partial}{\partial v}\, q1(u, v)\right) - 3 \left(\frac{\partial}{\partial v}\, q3(u, v)\right)}{u^2},$$
$$-\frac{1}{2}\, \frac{-2\, q4(u, v)\, u + 6 \left(\frac{\partial}{\partial v}\, q2(u, v)\right) u + 3\, q1(u, v) - 3\, q3(u, v)}{u^2},$$
$$\frac{1}{6}\, \frac{-2\, q4(u, v)\, u + 6 \left(\frac{\partial}{\partial v}\, q2(u, v)\right) u + 3\, q1(u, v) - 3\, q3(u, v)}{u},$$
$$\frac{3}{4}\, \frac{1}{u^3} \left(2 \left(\frac{\partial}{\partial u}\, q4(u, v)\right) u^2 - 4\, q4(u, v)\, u - 3 \left(\frac{\partial}{\partial u}\, q1(u, v)\right) u + 3 \left(\frac{\partial}{\partial u}\, q3(u,\right.\right.$$
$$\left.\left. v)\right) u + 9\, q1(u, v) - 9\, q3(u, v)\right), \frac{\partial}{\partial u}\, q1(u, v) - \frac{q1(u, v)}{u}, \frac{\partial}{\partial v}\, q1(u, v)$$
$$+ \frac{q2(u, v)}{u}, \frac{\partial}{\partial u}\, q2(u, v) - \frac{2\, q2(u, v)}{u}, \frac{\partial}{\partial v}\, q3(u, v) + \frac{q2(u, v)}{u}, \frac{\partial}{\partial u}\, q3(u, v)$$
$$\left. - \frac{q3(u, v)}{u} + \frac{2}{3}\, q4(u, v), \frac{\partial}{\partial v}\, q4(u, v) \right\}$$

How many equations are there?

```
N > nops(s3);
```

$$12 \tag{27}$$

This system may be easier to solve than the Killing equation itself. Let's get the solution.

```
N > s4 := pdsolve(s3, {seq(q||i(u,v),i=1..nops(rIT1))} );
```

$$s4 := \left\{q1(u, v) = \left(-\frac{1}{2}\_C1\,v^2 - \_C2\,v + \_C3\right)u,\ q2(u, v) = (\_C1\,v + \_C2)\,u^2,\right. \tag{28}$$
$$q3(u, v) = -\frac{1}{6}\,u\,(12\,\_C1\,u^2 + 3\,\_C1\,v^2 + 6\,\_C2\,v + 4\,\_C4\,u - 6\,\_C3),\ q4(u, v)$$
$$\left.= 6\,\_C1\,u^2 + \_C4\,u\right\}$$

How many independent solutions are there?

```
N > has(s4, _C4);
```

$$true \tag{29}$$

```
N > has(s4, _C5);
```

$$false \tag{30}$$

Thus, there are 4 independent solutions. Here is then what the parallel sections should look like:

```
N > s5 := DETools:-dsubs(s4, s1);
```

$$s5 := -\frac{(\_C1\,v^2 + 2\,\_C2\,v - 2\,\_C3)\,u}{2}\,E1 + (\_C1\,v + \_C2)\,u^2\,E2 \tag{31}$$
$$-\frac{u\,(12\,\_C1\,u^2 + 3\,\_C1\,v^2 + 6\,\_C2\,v + 4\,\_C4\,u - 6\,\_C3)}{6}\,E3 - 3\,u\,(\_C1\,v$$
$$+ \_C2)\,E4 + \left(6\,\_C1\,u^2 + \_C4\,u\right)E5 + \frac{9\,u\,\_C1}{2}\,E6$$

Now we will generate a list of parallel sections according to the independent solutions.

```
N > t1 := eval(s5, [_C1=1,_C2=0,_C3=0,_C4=0]);
    t2 := eval(s5, [_C1=0,_C2=1,_C3=0,_C4=0]);
    t3 := eval(s5, [_C1=0,_C2=0,_C3=1,_C4=0]);
    t4 := eval(s5, [_C1=0,_C2=0,_C3=0,_C4=1]);
```

$$t1 := -\frac{v^2\,u}{2}\,E1 + v\,u^2\,E2 - \frac{u\,(12\,u^2 + 3\,v^2)}{6}\,E3 - 3\,u\,v\,E4 + 6\,u^2\,E5 + \frac{9\,u}{2}\,E6$$

$$t2 := -u\,v\,E1 + u^2\,E2 - u\,v\,E3 - 3\,u\,E4 + 0\,E5 + 0\,E6$$

$$t3 := u\,E1 + 0\,E2 + u\,E3 + 0\,E4 + 0\,E5 + 0\,E6$$

$$t4 := 0\,E1 + 0\,E2 - \frac{2\,u^2}{3}\,E3 + 0\,E4 + u\,E5 + 0\,E6 \tag{32}$$

We check that they're parallel sections:

```
N > map(CovariantDerivative, [t1,t2,t3,t4], eval(C, p=1) );
```

$$[0\,E1 \otimes du,\ 0\,E1 \otimes du,\ 0\,E1 \otimes du,\ 0\,E1 \otimes du] \tag{33}$$

Now we drop these parallel sections down:

```
N > T := map(getHauserKT2, [t1,t2,t3,t4], N);
```

$$T := \left[-\frac{v^2\,u}{2}\,du \otimes du + v\,u^2\,du \otimes dv + v\,u^2\,dv \otimes du - \frac{u\,(4\,u^2+v^2)}{2}\,dv \otimes dv, -u\,v\,du \otimes du + u^2\,du \otimes dv + u^2\,dv \otimes du - u\,v\,dv \otimes dv,\ u\,du \otimes du + u\,dv \otimes dv,\ -\frac{2\,u^2}{3}\,dv \otimes dv\right] \tag{34}$$

Let's check that these are Killing tensors of rank 2.

```
N > map2(CheckKillingTensor, eval(G, p=1), T);
```

$$[0\,du \otimes du \otimes du,\ 0\,du \otimes du \otimes du,\ 0\,du \otimes du \otimes du,\ 0\,du \otimes du \otimes du] \tag{35}$$

Now let's check that we can lift the Killing tensors of the metric to parallel sections.
We begin by calculating the Killing tensors conventionally.

```
N > kt2 := KillingTensors(eval(G, p=1), 2);
```

$$kt2 := \left[\frac{v^2\,u}{2}\,du \otimes du - v\,u^2\,du \otimes dv - v\,u^2\,dv \otimes du + \left(\frac{1}{2}\,v^2\,u + 2\,u^3\right)\,dv \otimes dv,\ u\,v\,du \otimes du - u^2\,du \otimes dv - u^2\,dv \otimes du + u\,v\,dv \otimes dv,\ u\,du \otimes du + u\,dv \otimes dv,\ u^2\,dv \otimes dv\right] \tag{36}$$

Note that the Christoffel symbols of G are needed for the lift.

```
N > liftkt2 := map(HauserTractorLift2, kt2, eval(Gamma, p=1), N);
```

$$liftkt2 := \left[\frac{v^2\,u}{2}\,E1 - v\,u^2\,E2 + \left(\frac{1}{2}\,v^2\,u + 2\,u^3\right)\,E3 + 3\,u\,v\,E4 - 6\,u^2\,E5 - \frac{9\,u}{2}\,E6,\ u\,v\,E1 - u^2\,E2 + u\,v\,E3 + 3\,u\,E4,\ u\,E1 + u\,E3,\ u^2\,E3 - \frac{3\,u}{2}\,E5\right] \tag{37}$$

Let's check that these are parallel sections.

```
N > map(CovariantDerivative, liftkt2, eval(C, p=1) );
```

$$[0\, E1 \otimes du, 0\, E1 \otimes du, 0\, E1 \otimes du, 0\, E1 \otimes du] \tag{38}$$

But are there more parallel sections?

```
N > CovariantlyConstantTensors(eval(C, p=1), [seq(E||i,i=1..6)]);
```

$$\left[-\frac{v^2 u}{9} E1 + \frac{2\, v\, u^2}{9} E2 - \frac{u\,(4\, u^2 + v^2)}{9} E3 - \frac{2\, u\, v}{3} E4 + \frac{4\, u^2}{3} E5 + u\, E6, -\frac{2\, u^2}{3} E3 + u\, E5, \frac{u\, v}{3} E1 - \frac{u^2}{3} E2 + \frac{u\, v}{3} E3 + u\, E4, u\, E1 + u\, E3\right] \tag{39}$$

```
N > nops(%);
```

$$4 \tag{40}$$

Thus, the parallel sections and the Killing tensors are in one-to-one correspondence.