Appendix D: Software demonstration for Killing-Yano tensors.

In this demonstration, we will illustrate the utility of the programs with the following names: KillingTensorLibrary, MaxKY, BundleLift, KYTracCon, getKY, liftKY, and KYtoKT.

First, we will load in the required packages and read in the file which contains the programs we have written.

```
> read "TractorPrograms.txt";
  with(DifferentialGeometry):
  with(Tensor):
```

Now, we will read in an example metric in 3 dimensions. In using the KillingTensorLibrary command, we will initialize a coordinate environment M.

```
> g1 := KillingTensorLibrary(3, M);
```

$$g1 := dx \otimes dx + dy \otimes dy + x^{p}\, dz \otimes dz \tag{1}$$

The metric we will use will have p=1.

```
M > g := eval(g1, p=1);
```

$$g := dx \otimes dx + dy \otimes dy + x\, dz \otimes dz \tag{2}$$

We need to know the required size of the fibers of the tractor bundle. We find this to be 10:

```
M > MaxKY(3, 2);
```

$$4 \tag{3}$$

Now we initialize the required environments.

```
M > DGEnvironment[VectorSpace](4, V);
```

Vector Space: V **(4)**

```
V > DGEnvironment[VectorBundle](M, V, N);
```

Vector Bundle: N **(5)**

Now that the vector bundle has been initialized, it is convenient to redefine the metric on this bundle:

```
N > G := BundleLift(g, N);
```

(6)

$$G := dx \otimes dx + dy \otimes dy + x\, dz \otimes dz \tag{6}$$

We also need the Christoffel symbols.

```
N > Gamma := Christoffel(G);
```

$$\Gamma := \nabla_{\partial_x} \partial_z = \frac{1}{2\,x}\,\partial_z,\ \nabla_{\partial_z} \partial_x = \frac{1}{2\,x}\,\partial_z,\ \nabla_{\partial_z} \partial_z = -\left(\frac{1}{2}\right)\partial_x \tag{7}$$

Now we can compute the tractor connection. Note that the rank of the Killing-Yano tensor must be specified.

```
N > C := KYTracCon(Gamma, 2, N);
```

$$C := \nabla_{\partial_x} E2 = -\frac{1}{2\,x}\,E2,\ \nabla_{\partial_x} E3 = -\frac{1}{2\,x}\,E3 - \frac{1}{8\,x^2}\,E4,\ \nabla_{\partial_x} E4 = E3 - \frac{1}{2\,x}\,E4,\ \nabla_{\partial_y} E4 = -E2,\ \nabla_{\partial_z} E1 = -\left(\frac{1}{2}\right) E3 - \frac{1}{8\,x}\,E4,\ \nabla_{\partial_z} E3 = \frac{1}{2\,x}\,E1,\ \nabla_{\partial_z} E4 = E1 \tag{8}$$

We now compute the Curvature tensor for the tractor connection.

```
N > K := CurvatureTensor(C);
```

$$K := -\frac{1}{8\,x^2}\,E1 \otimes \Theta3 \otimes dx \otimes dz + \frac{1}{8\,x^2}\,E1 \otimes \Theta3 \otimes dz \otimes dx + \frac{1}{8\,x}\,E2 \otimes \Theta1 \otimes dy \otimes dz - \frac{1}{8\,x}\,E2 \otimes \Theta1 \otimes dz \otimes dy - \frac{1}{8\,x^2}\,E2 \otimes \Theta3 \otimes dx \otimes dy + \frac{1}{8\,x^2}\,E2 \otimes \Theta3 \otimes dy \otimes dx + \frac{1}{8\,x}\,E3 \otimes \Theta1 \otimes dx \otimes dz - \frac{1}{8\,x}\,E3 \otimes \Theta1 \otimes dz \otimes dx + \frac{1}{4\,x^2}\,E4 \otimes \Theta1 \otimes dx \otimes dz - \frac{1}{4\,x^2}\,E4 \otimes \Theta1 \otimes dz \otimes dx \tag{9}$$

We can represent this tensor as a collection of (1,1) tensors by contracting the curvature tensor with the coordinate vectors of the base manifold.

```
N > k1 := ContractIndices(K, evalDG(D_x &t D_y), [[3,1],[4,2]]);
```

$$k1 := -\frac{1}{8\,x^2}\,E2 \otimes \Theta3 \tag{10}$$

```
N > k2 := ContractIndices(K, evalDG(D_x &t D_z), [[3,1],[4,2]]);
```

$$k2 := -\frac{1}{8\,x^2}\,E1 \otimes \Theta3 + \frac{1}{8\,x}\,E3 \otimes \Theta1 + \frac{1}{4\,x^2}\,E4 \otimes \Theta1 \tag{11}$$

```
N > k3 := ContractIndices(K, evalDG(D_y &t D_z), [[3,1],[4,2]]);
```

$$k3 := \frac{1}{8\,x}\,E2 \otimes \Theta1 \tag{12}$$

We can think of these as matrices

```
N > K1 := convert(k1, DGMatrix);
    K2 := convert(k2, DGMatrix);
    K3 := convert(k3, DGMatrix);
```

$$K1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & -\frac{1}{8x^2} & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

$$K2 := \begin{bmatrix} 0 & 0 & -\frac{1}{8x^2} & 0 \\ 0 & 0 & 0 & 0 \\ \frac{1}{8x} & 0 & 0 & 0 \\ \frac{1}{4x^2} & 0 & 0 & 0 \end{bmatrix}$$

$$K3 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ \frac{1}{8x} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{13}$$

and subsequently find the basis of the 0th order reduced tractor bundle:

```
N > IT0 := LieAlgebras:-InvariantTensors([K1,K2,K3], [seq(E||i,i=
    1..4)]);
```

$$IT0 := [E2, E4] \tag{14}$$

How many are there?

```
N > nops(IT0);
```

$$2 \tag{15}$$

Thus, there are a maximum of 2 rank 2 Killing Yano tensors.

Now let's see if we can get the Killing-Yano tensors explicitly. We begin by forming a function-coefficient linear combination of the basis elements.

```
N > s1 := DGzip([seq(q||i(x,y,z), i=1..nops(IT0))], IT0, "plus");
```

$$s1 := q1(x, y, z)\, E2 + q2(x, y, z)\, E4 \quad (16)$$

Next, we take the covariant derivative.

```
N > s2 := CovariantDerivative(s1, C);
```

$$s2 := q2(x, y, z)\, E1 \otimes dz + \left(\frac{\partial}{\partial x}\, q1(x, y, z) - \frac{q1(x, y, z)}{2\,x}\right) E2 \otimes dx + \left(\frac{\partial}{\partial y}\, q1(x, y, z) - q2(x, y, z)\right) E2 \otimes dy + \frac{\partial}{\partial z}\, q1(x, y, z)\, E2 \otimes dz + q2(x, y, z)\, E3 \otimes dx + \left(\frac{\partial}{\partial x}\, q2(x, y, z) - \frac{q2(x, y, z)}{2\,x}\right) E4 \otimes dx + \frac{\partial}{\partial y}\, q2(x, y, z)\, E4 \otimes dy + \frac{\partial}{\partial z}\, q2(x, y, z)\, E4 \otimes dz \quad (17)$$

Let's look at the equations we need to solve.

```
N > s3 := DGinformation(s2, "CoefficientSet");
```

$$s3 := \left\{\frac{\partial}{\partial x}\, q1(x, y, z) - \frac{1}{2}\,\frac{q1(x, y, z)}{x}, \frac{\partial}{\partial y}\, q1(x, y, z) - q2(x, y, z), \frac{\partial}{\partial x}\, q2(x, y, z) - \frac{1}{2}\,\frac{q2(x, y, z)}{x}, \frac{\partial}{\partial z}\, q1(x, y, z), \frac{\partial}{\partial y}\, q2(x, y, z), \frac{\partial}{\partial z}\, q2(x, y, z), q2(x, y, z)\right\} \quad (18)$$

How many equations are there?

```
N > nops(s3);
```

$$7 \quad (19)$$

This system may be easier to solve than the Killing equation itself. Let's get the solution.

```
N > s4 := pdsolve(s3, {seq(q||i(x,y,z),i=1..nops(IT0))} );
```

$$s4 := \left\{q1(x, y, z) = \_C1\sqrt{x}, q2(x, y, z) = 0\right\} \quad (20)$$

How many independent solutions are there?

```
N > has(s4, _C1);
```

true **(21)**

```
N > has(s4, _C2);
```

false **(22)**

Thus, there is 1 independent solution and, consequently, a single Killing-Yano tensor. Here is then what the parallel section should look like:

```
N > s5 := DETools:-dsubs(s4, s1);
```

$$s5 := \_C1\sqrt{x}\ E2 + 0\ E4 \tag{23}$$

We can evaluate this at _C1=1.

```
N > t1 := eval(s5, [_C1=1]);
```

$$t1 := \sqrt{x}\ E2 + 0\ E4 \tag{24}$$

We check that it's a parallel section:

```
N > CovariantDerivative(t1, C);
```

$$0\ E1 \otimes dx \tag{25}$$

Now we drop this parallel section down. Note that the rank of the Killing-Yano tensor must again be specified.

```
N > T := getKY(t1, 2, N);
```

$$T := \sqrt{x}\ dx \wedge dz \tag{26}$$

Now let's check that we can lift the Killing-Yano tensors of the metric to parallel sections. We begin by calculating them conventionally.

```
N > ky := KillingYanoTensors(G, 2);
```

$$ky := \left[\sqrt{x}\ dx \wedge dz\right] \tag{27}$$

Now let's lift it to a section. The rank is not required here.

```
N > liftky := liftKY(ky[1], Gamma, N);
```

$$liftky := \sqrt{x}\ E2 \tag{28}$$

Let's check that this is a parallel section.

```
N > CovariantDerivative(liftky, C);
```

$$0\ E1 \otimes dx \tag{29}$$

But are there more parallel sections?

```
N > CovariantlyConstantTensors(C, [seq(E||i,i=1..4)]);
```

$$\left[\sqrt{x}\ E2\right] \tag{30}$$

```
N > nops(%);
```

$$1 \tag{31}$$

Thus, the parallel sections and the Killing tensors are in one-to-one correspondence.

We will now construct a Killing tensor of rank 2 from the known Killing-Yano tensor.

```
N > KT := KYtoKT(G, InverseMetric(G), ky[1], ky[1]);
```

$$KT := -dx \otimes dx - x\, dz \otimes dz \tag{32}$$

Let's check that this is a Killing tensor of G.

```
N > CheckKillingTensor(G,KT);
```

$$0\, dx \otimes dx \otimes dx \tag{33}$$

The KillingTensorLibrary command can also be used to call other known quantities of a metric, such as the Killing tensors of rank 1:

```
N > kt1 := eval(map(BundleLift, KillingTensorLibrary(3, M,
    output=["KillingTensors", 1]), N), p=1);
```

$$kt1 := [x\, dz, dy] \tag{34}$$

Now, let's get a basis for the space of known Killing tensors, including the metric. In principle, this can be done using the tractor approach, but a more conventional command exists and alliviates the need to construct the tractor bundle for Killing tensors of rank 2.

```
N > reds := SymmetricProductsOfKillingTensors([kt1, [G]], 2);
```

$$reds := \left[x^2\, dz \otimes dz, \frac{x}{2}\, dy \otimes dz + \frac{x}{2}\, dz \otimes dy, dy \otimes dy, dx \otimes dx + dy \otimes dy + x\, dz \otimes dz\right] \tag{35}$$

Now we will determine, conventionally, whether the Killing tensor we've newly constructed is a linear combination of the known Killing tensors.

```
N > GetComponents(KT, reds);
```

$$[0, 0, 1, -1] \tag{36}$$

Thus, the Killing tensor so constructed is a linear combination of the Killing tensors which are already known. It is, however, not a linear combination of only the reducible Killing tensors, and is therefore irreducible--the metric itself is irreducible, in this case.

For the sake of curiosity, and since irreducible Killing tensors are of such interest, we will calculate all Killing tensors of rank 2 for the metric G. In principle, this can be done by means of the tractor approach; however, this particular metric presents no obstacles in computing the Killing tensors directly:

```
N > kt2 := KillingTensors(G, 2);
```

$$kt2 := \left[y\, dx \otimes dx - \frac{x}{2}\, dx \otimes dy - \frac{x}{2}\, dy \otimes dx - \frac{z\, x}{4}\, dy \otimes dz - \frac{z\, x}{4}\, dz \otimes dy + x\, y\, dz \otimes dz,\ dx \otimes dx + x\, dz \otimes dz,\ x^2\, dz \otimes dz,\ \frac{x}{2}\, dy \otimes dz + \frac{x}{2}\, dz \otimes dy,\ dy \otimes dy\right] \quad (37)$$

We now find which Killing tensors, if any, are not linear combinations of the metric and of the reducible Killing tensors.

```
N > irreds := IndependentKillingTensors(kt2, reds);
```

$$irreds := \left[y\, dx \otimes dx - \frac{x}{2}\, dx \otimes dy - \frac{x}{2}\, dy \otimes dx - \frac{z\, x}{4}\, dy \otimes dz - \frac{z\, x}{4}\, dz \otimes dy + x\, y\, dz \otimes dz\right] \quad (38)$$

Thus, the Killing tensor above is not a linear combination of the metric and the reducible Killing tensors. In particular, it is irreducible.