# Appendix A: Software demonstration for Killing vectors

In this demonstration, we will illustrate the utility of the programs with the following names: MaxKT, BundleLift, rnk1TracConn, getRnk1, and liftrnk1.

First, we will read in the file which contains the programs we have written and load in other necessary packages.

```
> read "TractorPrograms.txt";
  with(DifferentialGeometry):
  with(Tensor):
```

Now, we will initialize a 2-dimensional manifold.

```
> DGEnvironment[Coordinate]([u,v], M);
```

*Manifold: M* **(1)**

Next, we will define a simple metric.

```
M > g := evalDG( u^p * (du &t du + dv &t dv) );
```

$$g := u^p\, du \otimes du + u^p\, dv \otimes dv \quad \textbf{(2)}$$

We need to know the required size of the fibers of the tractor bundle. We find this to be 3:

```
M > MaxKT(2, 1);
```

3 **(3)**

Now we initialize the required environments.

```
M > DGEnvironment[VectorSpace](3, V);
```

*Vector Space: V* **(4)**

```
V > DGEnvironment[VectorBundle](M, V, N);
```

*Vector Bundle: N* **(5)**

Now that the vector bundle has been initialized, it is convenient to redefine the metric on this bundle:

```
N > G := BundleLift(g, N);
```

$$G := u^p\, du \otimes du + u^p\, dv \otimes dv \quad \textbf{(6)}$$

We will also need the Christoffel symbols:

```
N > Gamma := Christoffel(G);
```

$$\Gamma := \nabla_{\partial_u} \partial_u = \frac{p}{2\,u}\,\partial_u, \nabla_{\partial_u} \partial_v = \frac{p}{2\,u}\,\partial_v, \nabla_{\partial_v} \partial_u = \frac{p}{2\,u}\,\partial_v, \nabla_{\partial_v} \partial_v = -\frac{p}{2\,u}\,\partial_u \quad (7)$$

Now we can compute the tractor connection:

```
N > C := rnk1TracConn(Gamma, N);
```

$$C := \nabla_{\partial_u} E1 = -\frac{p}{2\,u}\,E1, \nabla_{\partial_u} E2 = -\frac{p}{2\,u}\,E2 + \frac{p}{2\,u^2}\,E3, \nabla_{\partial_u} E3 = -E2 - \frac{p}{u}\,E3, \nabla_{\partial_v} E1 = \frac{p}{2\,u}\,E2 - \frac{p}{2\,u^2}\,E3, \nabla_{\partial_v} E2 = -\frac{p}{2\,u}\,E1, \nabla_{\partial_v} E3 = E1 \quad (8)$$

We can also represent this connection as a matrix of 1-forms:

```
N > convert(C, DGMatrix);
```

$$\begin{bmatrix} -\frac{p}{2\,u}\,du & -\frac{p}{2\,u}\,dv & dv \\ \frac{p}{2\,u}\,dv & -\frac{p}{2\,u}\,du & -du \\ -\frac{p}{2\,u^2}\,dv & \frac{p}{2\,u^2}\,du & -\frac{p}{u}\,du \end{bmatrix} \quad (9)$$

We now compute the Curvature tensor for the tractor connection.

```
N > K := CurvatureTensor(C);
```

$$K := \frac{p\,(2+p)}{2\,u^3}\,E3 \otimes \Theta1 \otimes du \otimes dv - \frac{p\,(2+p)}{2\,u^3}\,E3 \otimes \Theta1 \otimes dv \otimes du \quad (10)$$

It is interesting to note that p=0 and p=-2 are the only values of p for which the curvature tensor vanishes identically.

We can represent this tensor as a collection of (1,1) tensors by contracting the curvature tensor with the coordinate vectors of the base manifold. In our 2-dimensional case, there is only one such (1,1) tensor, but there may be others in higher dimensions.

```
N > k1 := ContractIndices(K, evalDG(D_u &t D_v), [[3,1],[4,2]] );
```

$$k1 := \frac{p\,(2+p)}{2\,u^3}\,E3 \otimes \Theta1 \quad (11)$$

We can think of this as a matrix

```
N > K1 := convert(k1, DGMatrix);
```

$$K1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ \frac{1}{2}\,\frac{p\,(2+p)}{u^3} & 0 & 0 \end{bmatrix} \tag{12}$$

and subsequently find the basis for the 0th order reduced tractor bundle:

```
N > IT0 := LieAlgebras:-InvariantTensors([K1], [seq(E||i,i=1..3)]
    ) ;
```

$$IT0 := [E2, E3] \tag{13}$$

This calculation is assuming that p is not 0 or 2: caution is advised when examining metrics with unknown constants as exponents. Assuming that p is not 0 or 2, there are a maximum of 2 Killing vectors of the metric g, since there are 2 invariants.

We will now try to get a tighter upper bound using the 1st order reduced tractor bundle. We begin by differentiating the curvature tensor.

```
N > dK1 := CovariantDerivative(k1, C);
```

$$dK1 := \frac{p\,(2+p)}{2\,u^3}\,E1 \otimes \Theta 1 \otimes dv - \frac{p\,(2+p)}{2\,u^3}\,E2 \otimes \Theta 1 \otimes du - \frac{p\,(2+p)\,(6+p)}{4\,u^4}\,E3 \otimes \Theta 1 \otimes du + \frac{p^2\,(2+p)}{4\,u^4}\,E3 \otimes \Theta 2 \otimes dv - \frac{p\,(2+p)}{2\,u^3}\,E3 \otimes \Theta 3 \otimes dv \tag{14}$$

We now generate a set of (1,1) tensors by contraction:

```
N > D1K1 := ContractIndices(dK1, D_u, [[3,1]] );
```

$$D1K1 := -\frac{p\,(2+p)}{2\,u^3}\,E2 \otimes \Theta 1 - \frac{p\,(2+p)\,(6+p)}{4\,u^4}\,E3 \otimes \Theta 1 \tag{15}$$

```
N > D2K1 := ContractIndices(dK1, D_v, [[3,1]] );
```

$$D2K1 := \frac{p\,(2+p)}{2\,u^3}\,E1 \otimes \Theta 1 + \frac{p^2\,(2+p)}{4\,u^4}\,E3 \otimes \Theta 2 - \frac{p\,(2+p)}{2\,u^3}\,E3 \otimes \Theta 3 \tag{16}$$

Now we will convert them to matrices.

```
N > Md1k1 := convert(D1K1, DGMatrix);
```

$$Md1k1 := \begin{bmatrix} 0 & 0 & 0 \\ -\frac{1}{2}\,\frac{p\,(2+p)}{u^3} & 0 & 0 \\ -\frac{1}{4}\,\frac{p\,(2+p)\,(6+p)}{u^4} & 0 & 0 \end{bmatrix} \quad \textbf{(17)}$$

```
N > Md2k1 := convert(D2K1, DGMatrix);
```

$$Md2k1 := \begin{bmatrix} \frac{1}{2}\,\frac{p\,(2+p)}{u^3} & 0 & 0 \\ 0 & 0 & 0 \\ 0 & \frac{1}{4}\,\frac{p^2\,(2+p)}{u^4} & -\frac{1}{2}\,\frac{p\,(2+p)}{u^3} \end{bmatrix} \quad \textbf{(18)}$$

And now we can compute a basis for the 1st order reduced tractor connection.

```
N > IT1 := LieAlgebras:-InvariantTensors([K1,Md1k1,Md2k1], [seq
    (E||i,i=1..3)] );
```

$$IT1 := \left[E2 + \frac{p}{2\,u}\,E3\right] \quad \textbf{(19)}$$

Thus, we get a maximum of 1 Killing vector. Let's be sure that our calculation doesn't depend on p not being 6, since p=6 appears, from looking at the matrices above, to be an exceptional value.

```
N > LieAlgebras:-InvariantTensors( eval([K1,Md1k1,Md2k1], p=6),
    [seq(E||i,i=1..3)]);
```

$$\left[E2 + \frac{3}{u}\,E3\right] \quad \textbf{(20)}$$

Now let's see if we can get the Killing vector explicitly. First, we form a linear combination of the 1st order reduced tractor basis elements with function coefficients.

```
N > s1 := DGzip([seq(q||i(u,v), i=1..nops(IT1))] ,IT1, "plus");
```

$$s1 := q1(u,v)\,E2 + \frac{q1(u,v)\,p}{2\,u}\,E3 \quad \textbf{(21)}$$

Next, we take the covariant derivative.

```
N > s2 := CovariantDerivative(s1, C);
```

**(22)**

$$s2 := -\left(-\left(\frac{\partial}{\partial u}\, q1(u,v)\right) + \frac{q1(u,v)\,p}{u}\right) E2 \otimes du + \frac{\partial}{\partial v}\, q1(u,v)\, E2 \otimes dv + \frac{p\left(\left(\frac{\partial}{\partial u}\, q1(u,v)\right) u - q1(u,v)\,p\right)}{2\,u^2}\, E3 \otimes du + \frac{\left(\frac{\partial}{\partial v}\, q1(u,v)\right) p}{2\,u}\, E3 \otimes dv \quad \textbf{(22)}$$

Let's look at the equations we need to solve.

```
N > s3 := DGinformation(s2, "CoefficientSet");
```

$$s3 := \left\{ \frac{1}{2}\, \frac{p\left(\left(\frac{\partial}{\partial u}\, q1(u,v)\right) u - q1(u,v)\,p\right)}{u^2},\ \frac{1}{2}\, \frac{\left(\frac{\partial}{\partial v}\, q1(u,v)\right) p}{u},\ \frac{\partial}{\partial u}\, q1(u,v) - \frac{q1(u,v)\,p}{u},\ \frac{\partial}{\partial v}\, q1(u,v) \right\} \quad \textbf{(23)}$$

How many equations are there?

```
N > nops(s3);
```

$$4 \quad \textbf{(24)}$$

This is an easier system to solve than the Killing equations themselves, which has 3 equations and 2 unknown functions.

Let's get the solution:

```
N > s4 := pdsolve(s3, q||1(u,v));
```

$$s4 := \{q1(u,v) = \_C1\, u^p\} \quad \textbf{(25)}$$

Here is then what the parallel sections should look like:

```
N > s5 := DETools:-dsubs(s4, s1);
```

$$s5 := \_C1\, u^p\, E2 + \frac{\_C1\, u^p\, p}{2\,u}\, E3 \quad \textbf{(26)}$$

We may as well evaluate this at _C1=1.

```
N > s6 := eval(s5,_C1=1);
```

$$s6 := u^p\, E2 + \frac{u^p\, p}{2\,u}\, E3 \quad \textbf{(27)}$$

We check that it's a parallel section:

```
N > CovariantDerivative(s6, C);
```

$$0\, E1 \otimes du \tag{28}$$

Now we drop this parallel section down:

```
N > T := getRnk1(s6, N);
```

$$T := u^p\, dv \tag{29}$$

Let's check that this is a Killing tensor of rank 1.

```
N > CheckKillingTensor(G, T);
```

$$0\, du \otimes du \tag{30}$$

If the contravariant vector is desired, we can simply raise indices:

```
N > X := RaiseLowerIndices( InverseMetric(G), T, [1] );
```

$$X := \partial_v \tag{31}$$

Now let's check that we can lift the Killing tensors of the metric to parallel sections. We begin by calculating the Killing tensors conventionally.

```
N > kt1 := KillingTensors(G, 1);
```

$$kt1 := [u^p\, dv] \tag{32}$$

As there is only one, we will lift this individually rather than lift the list. Note that the Christoffel symbols of G are needed.

```
N > liftkt1 := liftrnk1(kt1[1], Gamma, N);
```

$$liftkt1 := u^p\, E2 + \frac{u^{p-1}\, p}{2}\, E3 \tag{33}$$

Let's check that this is a parallel section.

```
N > CovariantDerivative(liftkt1, C);
```

$$0\, E1 \otimes du \tag{34}$$

But are there more parallel sections?

```
N > CovariantlyConstantTensors(C, [seq(E||i,i=1..3)] );
```

$$\left[\frac{2\, u^p}{p}\, E2 + u^{p-1}\, E3\right] \tag{35}$$

Since there is only 1, which corresponds with our known parallel section, the parallel sections and the Killing vectors are in one-to-one correspondence.