Appendix B: Software demo for conformal Killing vectors.

In this demonstration, we will illustrate the utility of the programs with the following names: MaxCF, BundleLift, ConfTracConn, getConfKV, and liftConfKV.

First, we will read in the file which contains the programs we have written and load in other required packages.

```
> read "TractorPrograms.txt";
  with(DifferentialGeometry):
  with(Tensor):
```

Now, we will initialize a 3-dimensional manifold.

```
> DGEnvironment[Coordinate]([x,y,z], M);
```

Manifold: M **(1)**

Next, we will define a simple metric.

```
M > g := evalDG(dx &t dx + dy &t dy + x * dz &t dz);
```

$$g := dx \otimes dx + dy \otimes dy + x\, dz \otimes dz \tag{2}$$

We need to know the required size of the fibers of the tractor bundle. We find this to be 10:

```
M > MaxCF(3, 1);
```

10 **(3)**

Now we initialize the required environments.

```
M > DGEnvironment[VectorSpace](10, V);
```

Vector Space: V **(4)**

```
V > DGEnvironment[VectorBundle](M, V, N);
```

Vector Bundle: N **(5)**

Now that the vector bundle has been initialized, it is convenient to redefine the metric on this bundle:

```
N > G := BundleLift(g, N);
```

$$G := dx \otimes dx + dy \otimes dy + x\, dz \otimes dz \tag{6}$$

Now we can compute the tractor connection:

```
N > C := ConfTracConn(G, N);
```

$$C := \nabla_{\partial_x} E1 = -\frac{1}{4x^3}\,E8, \nabla_{\partial_x} E3 = -\frac{1}{2x}\,E3 + \frac{1}{4x^2}\,E5, \nabla_{\partial_x} E4 = -E2 - \frac{1}{4x^2}\,E9, \nabla_{\partial_x} E5 = -E3 - \frac{1}{2x}\,E5, \nabla_{\partial_x} E6 = -\frac{1}{2x}\,E6, \nabla_{\partial_x} E7 = E1 - \frac{1}{4x^2}\,E8, \nabla_{\partial_x} E8 = E7, \nabla_{\partial_x} E9 = E4, \nabla_{\partial_x} E10 = E5 - \frac{1}{2x}\,E10, \nabla_{\partial_y} E1 = \frac{1}{4x^3}\,E9, \nabla_{\partial_y} E4 = E1 - \frac{1}{4x^2}\,E8, \nabla_{\partial_y} E6 = -E3 + \frac{1}{4x^2}\,E10, \nabla_{\partial_y} E7 = E2 + \frac{1}{4x^2}\,E9, \nabla_{\partial_y} E8 = -E4, \nabla_{\partial_y} E9 = E7, \nabla_{\partial_y} E10 = E6, \nabla_{\partial_z} E1 = \left(\frac{1}{2}\right) E3 - \frac{1}{4x}\,E5 - \frac{1}{4x^2}\,E10, \nabla_{\partial_z} E3 = -\frac{1}{2x}\,E1, \nabla_{\partial_z} E4 = -\left(\frac{1}{2}\right) E6, \nabla_{\partial_z} E5 = E1, \nabla_{\partial_z} E6 = E2 + \frac{1}{2x}\,E4 + \frac{1}{4x^2}\,E9, \nabla_{\partial_z} E7 = x\,E3 - \frac{1}{4x}\,E10, \nabla_{\partial_z} E8 = -x\,E5 + \left(\frac{1}{2}\right) E10, \nabla_{\partial_z} E9 = -x\,E6, \nabla_{\partial_z} E10 = E7 - \frac{1}{2x}\,E8 \tag{7}$$

We now compute the Curvature tensor for the tractor connection.

```
N > K := CurvatureTensor(C);
```

$$K := \frac{1}{8x^4}\,E8\otimes\Theta3\otimes dx\otimes dz - \frac{1}{8x^4}\,E8\otimes\Theta3\otimes dz\otimes dx + \frac{1}{4x^3}\,E8\otimes\Theta4\otimes dx\otimes dy - \frac{1}{4x^3}\,E8\otimes\Theta4\otimes dy\otimes dx - \frac{1}{4x^3}\,E8\otimes\Theta5\otimes dx\otimes dz + \frac{1}{4x^3}\,E8\otimes\Theta5\otimes dz\otimes dx - \frac{3}{4x^4}\,E9\otimes\Theta1\otimes dx\otimes dy + \frac{3}{4x^4}\,E9\otimes\Theta1\otimes dy\otimes dx - \frac{1}{8x^4}\,E9\otimes\Theta3\otimes dy\otimes dz + \frac{1}{8x^4}\,E9\otimes\Theta3\otimes dz\otimes dy + \frac{1}{4x^3}\,E9\otimes\Theta5\otimes dy\otimes dz - \frac{1}{4x^3}\,E9\otimes\Theta5\otimes dz\otimes dy - \frac{1}{2x^3}\,E9\otimes\Theta6\otimes dx\otimes dz + \frac{1}{2x^3}\,E9\otimes\Theta6\otimes dz\otimes dx - \frac{3}{4x^3}\,E9\otimes\Theta7\otimes dx\otimes dy + \frac{3}{4x^3}\,E9\otimes\Theta7\otimes dy\otimes dx + \frac{3}{4x^3}\,E10\otimes\Theta1\otimes dx\otimes dz - \frac{3}{4x^3}\,E10\otimes\Theta1\otimes dz\otimes dx + \frac{1}{4x^2}\,E10\otimes\Theta4\otimes dy\otimes dz - \frac{1}{4x^2}\,E10\otimes\Theta4\otimes dz\otimes dy - \frac{1}{2x^3}\,E10\otimes\Theta6\otimes dx\otimes dy + \frac{1}{2x^3}\,E10\otimes\Theta6\otimes dy\otimes dx + \frac{3}{4x^2}\,E10\otimes\Theta7\otimes dx\otimes dz - \frac{3}{4x^2}\,E10\otimes\Theta7\otimes dz\otimes dx \tag{8}$$

We can represent this tensor as a collection of (1,1) tensors by contracting the curvature tensor with the coordinate vectors of the base manifold.

```
N > k1 := ContractIndices(K, evalDG(D_x &t D_y), [[3,1],[4,2]] );
```

$$k1 := \frac{1}{4\,x^3}\,E8 \otimes \Theta 4 - \frac{3}{4\,x^4}\,E9 \otimes \Theta 1 - \frac{3}{4\,x^3}\,E9 \otimes \Theta 7 - \frac{1}{2\,x^3}\,E10 \otimes \Theta 6 \qquad \textbf{(9)}$$

```
N > k2 := ContractIndices(K, evalDG(D_x &t D_z), [[3,1],[4,2]] );
```

$$k2 := \frac{1}{8\,x^4}\,E8 \otimes \Theta 3 - \frac{1}{4\,x^3}\,E8 \otimes \Theta 5 - \frac{1}{2\,x^3}\,E9 \otimes \Theta 6 + \frac{3}{4\,x^3}\,E10 \otimes \Theta 1 + \frac{3}{4\,x^2}\,E10 \otimes \Theta 7 \qquad \textbf{(10)}$$

```
N > k3 := ContractIndices(K, evalDG(D_y &t D_z), [[3,1],[4,2]] );
```

$$k3 := -\frac{1}{8\,x^4}\,E9 \otimes \Theta 3 + \frac{1}{4\,x^3}\,E9 \otimes \Theta 5 + \frac{1}{4\,x^2}\,E10 \otimes \Theta 4 \qquad \textbf{(11)}$$

We can think of these as matrices

```
N > K1 := convert(k1, DGMatrix);
    K2 := convert(k2, DGMatrix);
    K3 := convert(k3, DGMatrix);
```

$$K1 := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{4\,x^3} & 0 & 0 & 0 & 0 & 0 & 0 \\ -\frac{3}{4\,x^4} & 0 & 0 & 0 & 0 & 0 & -\frac{3}{4\,x^3} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{2\,x^3} & 0 & 0 & 0 & 0 \end{bmatrix}$$

$$K2 := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{8 x^4} & 0 & -\frac{1}{4 x^3} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{2 x^3} & 0 & 0 & 0 & 0 \\ \frac{3}{4 x^3} & 0 & 0 & 0 & 0 & 0 & \frac{3}{4 x^2} & 0 & 0 & 0 \end{bmatrix}$$

$$K3 := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -\frac{1}{8 x^4} & 0 & \frac{1}{4 x^3} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{1}{4 x^2} & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{12}$$

and subsequently find a basis for the 0th order reduced tractor bundle.

```
N > IT0 := LieAlgebras:-InvariantTensors([K1,K2,K3], [seq(E||i,i=
    1..10)]);
```

$$IT0 := \left[E1 - \frac{1}{x}\, E7, E2, E3 + \frac{1}{2\, x}\, E5, E8, E9, E10\right] \tag{13}$$

How many are there?

```
N > nops(IT0);
```

$$6 \tag{14}$$

Thus, there are a maximum of 6 conformal Killing vectors of the metric g.

Now let's see if we can get the Killing vectors explicitly. We begin by forming a function-coefficient linear combination of these basis elements.

```
N > s1 := DGzip([seq(q||i(x,y,z), i=1..nops(IT0))], IT0, "plus");
```

$$s1 := q1(x, y, z)\, E1 + q2(x, y, z)\, E2 + q3(x, y, z)\, E3 + \frac{q3(x, y, z)}{2\, x}\, E5 - \frac{q1(x, y, z)}{x}\, E7 + q4(x, y, z)\, E8 + q5(x, y, z)\, E9 + q6(x, y, z)\, E10 \tag{15}$$

Next, we take the covariant derivative.

```
N > s2 := CovariantDerivative(s1, C);
```

$$s2 := \left(\frac{\partial}{\partial x}\, q1(x, y, z) - \frac{q1(x, y, z)}{x}\right) E1 \otimes dx + \frac{\partial}{\partial y}\, q1(x, y, z)\, E1 \otimes dy + \frac{\partial}{\partial z}\, q1(x, y, z)\, E1 \otimes dz + \frac{\partial}{\partial x}\, q2(x, y, z)\, E2 \otimes dx + \left(\frac{\partial}{\partial y}\, q2(x, y, z) - \frac{q1(x, y, z)}{x}\right) E2 \otimes dy + \frac{\partial}{\partial z}\, q2(x, y, z)\, E2 \otimes dz + \left(\frac{\partial}{\partial x}\, q3(x, y, z) - \frac{q3(x, y, z)}{x}\right) E3 \otimes dx + \frac{\partial}{\partial y}\, q3(x, y, z)\, E3 \otimes dy + \left(\frac{\partial}{\partial z}\, q3(x, y, z) - \frac{q1(x, y, z)}{2}\right) E3 \otimes dz + q5(x, y, z)\, E4 \otimes dx - q4(x, y, z)\, E4 \otimes dy + \left(\frac{\frac{\left(\frac{\partial}{\partial x}\, q3(x, y, z)\right) x}{2} - \frac{q3(x, y, z)}{2}}{x^2} + q6(x, y, z)\right) E5 \otimes dx + \frac{\frac{\partial}{\partial y}\, q3(x, y, z)}{2\, x}\, E5 \otimes dy \tag{16}$$

$$
-\frac{4\,q4(x,y,z)\,x^2+q1(x,y,z)-2\left(\frac{\partial}{\partial z}\,q3(x,y,z)\right)}{4\,x}\,E5\otimes dz+q6(x,y,z)\,E6 \otimes dy-q5(x,y,z)\,x\,E6\otimes dz+\left(\frac{-\left(\frac{\partial}{\partial x}\,q1(x,y,z)\right)x+q1(x,y,z)}{x^2}+q4(x,y,z)\right)E7\otimes dx+\left(-\frac{\frac{\partial}{\partial y}\,q1(x,y,z)}{x}+q5(x,y,z)\right)E7\otimes dy+\left(-\frac{\frac{\partial}{\partial z}\,q1(x,y,z)}{x}+q6(x,y,z)\right)E7\otimes dz+\frac{\partial}{\partial x}\,q4(x,y,z)\,E8\otimes dx+\frac{\partial}{\partial y}\,q4(x,y,z)\,E8\otimes dy+\left(\frac{\partial}{\partial z}\,q4(x,y,z)-\frac{q6(x,y,z)}{2\,x}\right)E8\otimes dz+\frac{\partial}{\partial x}\,q5(x,y,z)\,E9\otimes dx+\frac{\partial}{\partial y}\,q5(x,y,z)\,E9 \otimes dy+\frac{\partial}{\partial z}\,q5(x,y,z)\,E9\otimes dz+\left(\frac{\partial}{\partial x}\,q6(x,y,z)-\frac{q6(x,y,z)}{2\,x}\right)E10\otimes dx+\frac{\partial}{\partial y}\,q6(x,y,z)\,E10\otimes dy+\left(\frac{\partial}{\partial z}\,q6(x,y,z)+\frac{q4(x,y,z)}{2}\right)E10\otimes dz
$$

Let's look at the equations we need to solve.

```
N > s3 := DGinformation(s2, "CoefficientSet");
```

$$
s3:=\left\{\frac{1}{2}\,\frac{\frac{\partial}{\partial y}\,q3(x,y,z)}{x},\ -\frac{1}{4}\,\frac{4\,q4(x,y,z)\,x^2+q1(x,y,z)-2\left(\frac{\partial}{\partial z}\,q3(x,y,z)\right)}{x},\ -q5(x,y,z)\,x,\ -q4(x,y,z),\ -\frac{\frac{\partial}{\partial y}\,q1(x,y,z)}{x}+q5(x,y,z),\ -\frac{\frac{\partial}{\partial z}\,q1(x,y,z)}{x}+q6(x,y,z),\ \frac{-\left(\frac{\partial}{\partial x}\,q1(x,y,z)\right)x+q1(x,y,z)}{x^2}+q4(x,y,z),\ \frac{\frac{1}{2}\left(\frac{\partial}{\partial x}\,q3(x,y,z)\right)x-\frac{1}{2}\,q3(x,y,z)}{x^2}+q6(x,y,z),\ \frac{\partial}{\partial x}\,q1(x,y,z)-\frac{q1(x,y,z)}{x},\ \frac{\partial}{\partial y}\,q2(x,y,z)-\frac{q1(x,y,z)}{x},\ \frac{\partial}{\partial x}\,q3(x,y,z)-\frac{q3(x,y,z)}{x},\ \frac{\partial}{\partial z}\,q3(x,y,z)-\frac{1}{2}\,q1(x,y,z),\ \frac{\partial}{\partial z}\,q4(x,y,z)-\frac{1}{2}\,\frac{q6(x,y,z)}{x},\ \frac{\partial}{\partial x}\,q6(x,y,z) \tag{17}
$$

$$-\frac{1}{2}\frac{q6(x,y,z)}{x}, \frac{\partial}{\partial z}q6(x,y,z)+\frac{1}{2}q4(x,y,z), \frac{\partial}{\partial y}q1(x,y,z), \frac{\partial}{\partial z}q1(x,y,z), \frac{\partial}{\partial x}q2(x,y,z), \frac{\partial}{\partial z}q2(x,y,z), \frac{\partial}{\partial y}q3(x,y,z), \frac{\partial}{\partial x}q4(x,y,z), \frac{\partial}{\partial y}q4(x,y,z), \frac{\partial}{\partial x}q5(x,y,z), \frac{\partial}{\partial y}q5(x,y,z), \frac{\partial}{\partial z}q5(x,y,z), \frac{\partial}{\partial y}q6(x,y,z), q5(x,y,z), q6(x,y,z)\Big\}$$

How many equations are there?

```
N > nops(s3);
```

$$28 \tag{18}$$

This system may be easier to solve than the Killing equation itself. Let's get the solution.

```
N > s4 := pdsolve(s3, {seq(q||i(x,y,z), i=1..nops(IT0))} );
```

$$s4 := \{q1(x,y,z)=2\_C1\,x, q2(x,y,z)=2\_C1\,y+\_C3, q3(x,y,z)=(\_C1\,z+\_C2)\,x, q4(x,y,z)=0, q5(x,y,z)=0, q6(x,y,z)=0\} \tag{19}$$

How many independent solutions are there?

```
N > has(s4,_C3);
```

$$true \tag{20}$$

```
N > has(s4,_C4);
```

$$false \tag{21}$$

Thus, there are 3 independent solutions. Here is then what the parallel sections should look like:

```
N > s5 := DETools:-dsubs(s4, s1);
```

$$s5 := 2\_C1\,x\,E1+\left(2\_C1\,y+\_C3\right)E2+(\_C1\,z+\_C2)\,x\,E3+\left(\frac{\_C1\,z}{2}+\frac{\_C2}{2}\right)E5-2\_C1\,E7+0\,E8+0\,E9+0\,E10 \tag{22}$$

Now we will generate a list of parallel sections according to the independent solutions.

```
N > t1 := eval(s5, [_C1=1,_C2=0,_C3=0]);
    t2 := eval(s5, [_C1=0,_C2=1,_C3=0]);
    t3 := eval(s5, [_C1=0,_C2=0,_C3=1]);
```

$$t1 := 2\,x\,E1 + 2\,y\,E2 + z\,x\,E3 + \frac{z}{2}\,E5 - 2\,E7 + 0\,E8 + 0\,E9 + 0\,E10$$

$$t2 := 0\,E1 + 0\,E2 + x\,E3 + \left(\frac{1}{2}\right)E5 + 0\,E7 + 0\,E8 + 0\,E9 + 0\,E10$$

$$t3 := 0\,E1 + E2 + 0\,E3 + 0\,E5 + 0\,E7 + 0\,E8 + 0\,E9 + 0\,E10 \tag{23}$$

We check that they're parallel sections:

```
N > map(CovariantDerivative, [t1,t2,t3], C);
```

$$[0\,E1 \otimes dx,\ 0\,E1 \otimes dx,\ 0\,E1 \otimes dx] \tag{24}$$

Now we drop these parallel sections down to the base space:

```
N > T := map(getConfKV, [t1,t2,t3], N);
```

$$T := [2\,x\,dx + 2\,y\,dy + z\,x\,dz,\ x\,dz,\ dy] \tag{25}$$

These are the (covariant) conformal Killing vectors of the metric. Now let's check that we can lift the conformal Killing vectors of the metric to parallel sections. We begin by calculating them conventionally.

```
N > ckv := ConformalKillingVectors(G);
```

$$ckv := \left[2\,x\,\partial_x + 2\,y\,\partial_y + z\,\partial_z\right],\ \left[\partial_z,\ \partial_y\right] \tag{26}$$

Let's get the covariant versions:

```
N > ckt := ListTools:-FlattenOnce([map2(RaiseLowerIndices, G, ckv
    [1], [1]), map2(RaiseLowerIndices, G, ckv[2], [1])] );
```

$$ckt := [2\,x\,dx + 2\,y\,dy + z\,x\,dz,\ x\,dz,\ dy] \tag{27}$$

Now let's lift them to sections.

```
N > liftckt := map(liftConfKV, ckt, G, N);
```

$$liftckt := \left[2\,x\,E1 + 2\,y\,E2 + z\,x\,E3 + \frac{z}{2}\,E5 - 2\,E7,\ x\,E3 + \left(\frac{1}{2}\right)E5,\ E2\right] \tag{28}$$

Let's check that these are parallel sections.

```
N > map(CovariantDerivative, liftckt, C);
```

$$[0\,E1 \otimes dx,\ 0\,E1 \otimes dx,\ 0\,E1 \otimes dx] \tag{29}$$

But are there more parallel sections?

```
N > CovariantlyConstantTensors(C, [seq(E||i,i=1..10)]);
```

$$\left[-x\,E1-y\,E2-\frac{z\,x}{2}\,E3-\frac{z}{4}\,E5+E7,\;2\,x\,E3+E5,\;E2\right] \quad (30)$$

```
N  > nops(%);
```

$$3 \quad (31)$$

Thus, the parallel sections and the Killing tensors are in one-to-one correspondence.